AAEC/E 107

UNCLASSIFIED

AAEC/E107

AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT

LUCAS HEIGHTS

THEORY AND APPLICATION OF THE DOUBLE P_N METHOD IN SLAB GEOMETRY FOR ISOTROPIC NEUTRON SOURCES AND SCATTERING

by

J. J. THOMPSON *

* Attached, from The University of New South Wales

Issued Sydney, April 1963

UNCLASSIFIED

AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT

LUCAS HEIGHTS

THEORY AND APPLICATION OF THE DOUBLE P_N METHOD IN SLAB GEOMETRY FOR ISOTROPIC NEUTRON SOURCES AND SCATTERING

by

J. J. THOMPSON*

ABSTRACT

A simple matrix formalism is developed to facilitate the application of the double P_N method of spherical harmonics to multiple slab configurations, representative of some of the basic neutron transport problems in the theory of nuclear reactors. Special attention is given to the double P_2 and P_3 approximations, and the results of numerical computations are compared with known accurate results.

* Attached, from The University of New South Wales

CONTENTS

1. INTRODUCTION

A common problem in nuclear reactor analysis is the variation of the neutron scalar flux due to strong absorbers and boundaries. One approach to a solution is to isolate such problems, and then hope to apply the results of a detailed transport theory analysis to the usual few or multigroup diffusion equations which are necessarily inadequate for describing significant variations in angular flux. The transport theory results lead to parameters such as the linear extrapolation distance, extrapolated end point, self-shielding factor, etc., which are applied in normal diffusion theory by the assumption that the asymptotic transport theory and diffusion theory are the same.

For this reason at least, a reasonably powerful transport theory programme must be available. As a great deal can be learnt from one-dimensional slab problems with isotropic scattering, these problems constitute the first requirement for a transport theory programme.

In principle the usual spherical harmonics (P_N) method is capable of producing, in the limit, the exact result. In practice a large number of terms may be required when the angular flux varies rapidly or is discontinuous with angle. For such problems, which are characteristic of slab reactor calculations on fuel plates, control plates, etc., the double spherical harmonics (double P_N) method is preferred. Ziering and Schiff (1958) have described Yvon's method for slabs, but their formulation is not the most convenient for engineering reactor calculations. For multi-region problems their formulation leads to an unwieldy set of simultaneous equations.

This report presents a different approach, and illustrates the method by application to some classical problems, for example, Milne's problem. By the use of reflection and transmission matrices, application of the double P_N method to any problem in slab geometry requires (N + 1)'th order matrices only. Although the theory will be presented for any N, it appears that N = 2 or 3 will be adequate for most problems.

2. THEORY

It is assumed that the most general problem involves an array of slabs, each of different but uniform nuclear properties. In any one region an x co-ordinate can be measured in units of the local mean free path, and the monoenergetic neutron transport equation for isotropic scattering can be written:

$$\mu \frac{\partial}{\partial x} \phi(x,\mu) + \phi(x,\mu) = \frac{c}{2} \int_{-1}^{+1} \phi(x,\mu')\, d\mu' + \frac{1}{2} S(x) \quad , \qquad (1)$$

where c = number of secondaries per collision e.g., Σ_s/Σ_t , and S(x) = isotropic source strength per mean free path.

The double P_N spherical harmonics method is based on the expansion

$$\phi^{\pm}(x,\mu) = \sum_{n=o}^{\infty} (2n+1)\, B_n^{\pm}(x)\, P_n^{\pm}(\mu) \; . \qquad (2)$$

As μ is the cosine of the angle between the direction vector and the + ve x direction, it is convenient to define the half-range Legendre polynomials in a way slightly different from that of Ziering and Schiff, as

$$\left.\begin{aligned} P_n^{+}(\mu) &= P_n(2\mu - 1) \qquad (0 \le \mu < 1) \\ P_n^{-}(\mu) &= P_n(-2\mu - 1) \qquad (-1 \le \mu < 0) \end{aligned}\right\} \qquad (3)$$

The advantage is a clearer interpretation of B^+ and B^- that facilitates application of symmetry requirements and avoids excessive attention to co-ordinates.

With these definitions, the scalar flux is

$$\Phi(x) = \int_{-1}^{+1} \phi(x,\mu)\, d\mu = B_o^{+}(x) + B_o^{-}(x) \; . \qquad (4)$$

Similarly the components of the current in the + ve and - ve x directions are

$$J^+(x) = \frac{1}{2}\left(B_0^+(x) + B_1^+(x)\right)$$

$$J^-(x) = \frac{1}{2}\left(B_0^-(x) + B_1^-(x)\right) , \qquad (5)$$

giving a net neutron current

$$J(x) = J^+(x) - J^-(x) = \frac{1}{2}\left(B^+(x) - B^-(x)\right)_{1+2} . \qquad (6)$$

Substitution of (2) into (1), and application of the orthogonal properties of the Legendre polynomials leads to two sets of coupled equations for the flux components, which are conveniently written in matrix form.

Let C be a diagonal matrix with $C_{11} = c$, $C_{ii} = 0$, $i \neq 1$, let P be a diagonal matrix $\left(2-c, 6, 10, 14, \ldots\ldots 2(2i-1), \ldots\ldots\right)$ and let δ be the vector (1,0,0,0, ... 0, ...).

Also let B^+ be the vector (B_0^+, B_1^+, B_2^+ .... B_{i-1}^+, ...) with a similar definition for B^-, and let M be the following square matrix,

$$M = \begin{bmatrix} 1 & 1 & & & & & & \\ 1 & 3 & 2 & & & & & \\ & 2 & 5 & 3 & & & & \\ & & 3 & 7 & 4 & & & \\ & & & . & . & . & & \\ & & & & i-1 & 2i-1 & i & \\ & & & & & . & . & . \end{bmatrix} . \qquad (7)$$

Then, with $D = \frac{d}{dx}$, the equations are

$$\left.\begin{aligned} (DM + P)\,B^+ &= C\,B^- + S\,\delta \\ (-DM + P)\,B^- &= C\,B^+ + S\,\delta \end{aligned}\right\} . \qquad (8)$$

The first two equations give the neutron balance equations

$$2\frac{d}{dx} J^+(x) + (2-c)\,B_0^+ - c\,B_0^- = S , \qquad (9a)$$

$$-2\frac{d}{dx} J^-(x) + (2-c)\,B_0^- - c\,B_0^+ = S , \qquad (9b)$$

that is,

$$\frac{d}{dx} J(x) + (1-c)\,\Phi(x) = S(x) . \qquad (9c)$$

The solution for a uniform source in an infinite medium is therefore

$$B_0^+ = B_0^- = \frac{S}{2(1-c)} , \quad B_n^\pm = 0 , \quad n \neq 0 , \qquad (10)$$

which can be written as

$$B^+ = B^- = \frac{S}{2(1-c)}\,\delta . \qquad (11)$$

If solutions to (8) are sought in the form

$$B^+ = A^+ e^{\alpha x} \quad , \quad B^- = A^- e^{\alpha x} \quad ,$$

then there results the characteristic equation

$$| \alpha^2 M - (P + C) M^{-1} (P - C) | = 0 \ . \tag{12}$$

The matrix $M^{-1}(P + C)M^{-1}(P - C)$ is symmetric giving real roots. For $c < 1$ these are positive; for $c = 1$, there is a root $\alpha^2 = 0$; for $c > 1$, the lowest α^2 is negative. The double P_N approximation truncates the matrix, corresponding to

$$B_{N+1, 2, \ldots\ldots} = 0 \quad ,$$

and the characteristic equation becomes an $(N+1)$'th order polynomial in α^2. For $c \neq 0$, the general source free solution may be written

$$B^+ = X \phi^+ A + Y \phi^- a \tag{13a}$$

$$B^- = Y F \phi^+ A + X F^{-1} \phi^- a \ , \tag{13b}$$

where ϕ^+ and ϕ^- are the diagonal matrices,

$$(e^{\alpha_1 x} \quad e^{\alpha_2 x} \ldots e^{\alpha_{N+1} x}) \ , \quad (e^{-\alpha_1 x} \quad e^{-\alpha_2 x} \ldots e^{-\alpha_{N+1} x}) \ ,$$

while A and a are the unknown vectors. The square matrices X and Y and the diagonal matrix F will be given later for particular values of N.

For $c = 0$, the B^- and B^+ vectors are uncoupled as is necessary for a pure absorber, so for a source free medium the solutions are

$$B^+ = Y \phi^- a \tag{14a}$$

$$B^- = Y \phi^+ a \ . \tag{14b}$$

Some important results follow from these general solutions. Let $B^+ = X \phi^+ A$, $B^- = YF \phi^- A$, and let α be the diagonal matrix of the roots, arranged in order of increasing magnitude. Then since

$$DM(B^- - B^+) = (P - C)(B^- + B^+) \quad ,$$

therefore

$$M(YF - X) \alpha \phi^+ A = (P - C)(YF + X) \phi^+ A \ ,$$

that is

$$M(YF - X) \alpha = (P - C)(YF + X) \ . \tag{15a}$$

Similarly

$$M(YF + X) \alpha = (P + C)(YF - X) \ . \tag{15b}$$

From (15a) and (15b)

$$(YF + X) \alpha^2 = M^{-1}(P + C) M^{-1}(P - C)(YF + X) \ , \tag{16}$$

so the columns of $(YF + X)$ are latent vectors of the matrix $M^{-1}(P + C)M^{-1}(P - C)$ with latent roots α_i^2. Thus the columns of $(YF + X)$ are orthogonal.

3. REFLECTION AND TRANSMISSION MATRICES

3.1 Infinite Half Plane $(0 \leq x < \infty)$

If there are no sources

$$B^+(x) = Y \phi^-(x) a \ , \qquad B^-(x) = XF^{-1} \phi^-(x) a \ .$$

Thus $$B^-(0) = \beta_\infty B^+(0) , \quad (17a)$$

where $$\beta_\infty = XF^{-1}Y^{-1} . \quad (17b)$$

Note also that for any x , $B^-(x) = \beta_\infty B^+(x)$,

and $$B^+(x) = \gamma_\infty(x) B^+(0) , \quad (18a)$$

where $$\gamma_\infty(x) = Y \phi^-(x) Y^{-1} . \quad (18b)$$

The matrices β_∞ and γ_∞ will be referred to as the infinite medium reflection and transmission matrices.

3.2 Finite Slab of Thickness t $(0 \leq x \leq t)$

Let neutrons be incident on the surface $x = 0$ only, with the angular distribution defined by the vector $B^+(0)$. A reflection matrix β and a transmission matrix γ may be defined by

$$B^-(0) = \beta B^+(0) \quad (19a)$$

$$B^+(t) = \gamma B^+(0). \quad (19b)$$

However these entities are simply related to the infinite medium matrices β_∞ and $\gamma_\infty(t)$ as follows:

The neutron distribution on the surfaces $x = 0$ and $x = t$, regarded as portion of an infinite medium, due to an incident distribution B^+ is

$$B^+(0) = B^+ , \qquad B^-(0) = \beta_\infty B^+ ,$$

$$B^+(t) = \gamma_\infty(t) B^+ , \qquad B^-(t) = \beta_\infty \gamma_\infty(t) B^+ .$$

To remove the distribution $B^-(t)$, superimpose a finite slab solution with the characteristics

$$B^+(0) = 0 , \qquad B^-(0) = -\gamma\, \beta_\infty \gamma_\infty(t) B^+ ,$$

$$B^+(t) = -\beta\, \beta_\infty \gamma_\infty(t) B^+ , \qquad B^-(t) = -\beta_\infty \gamma_\infty(t) B^+ .$$

It follows that

$$\beta = \beta_\infty - \gamma\, \beta_\infty \gamma_\infty(t) , \quad (20a)$$

$$\gamma = \gamma_\infty(t) - \beta\, \beta_\infty \gamma_\infty(t) , \quad (20b)$$

and therefore

$$\beta = \left(\beta_\infty - \gamma_\infty(t)\, \beta_\infty \gamma_\infty(t)\right) \left(I - \beta_\infty \gamma_\infty(t)\, \beta_\infty \gamma_\infty(t)\right)^{-1} \quad (21a)$$

$$\gamma = \left(\gamma_\infty(t) - \beta_\infty \beta_\infty \gamma_\infty(t)\right) \left(I - \beta_\infty \gamma_\infty(t)\, \beta_\infty \gamma_\infty(t)\right)^{-1} . \quad (21b)$$

The matrices β and γ may therefore be computed from the more basic entities β_∞ and $\gamma_\infty(t)$ which are given by (17b) and (18b). Note that

$$\beta_\infty \gamma_\infty(t) = XF^{-1} \phi^-(t) Y^{-1} . \quad (22)$$

For problems in which the flux distribution is symmetrical about the centre line $(x = 0)$ of the slab, such that $B^+(x) = B^-(-x)$, the matrix required is the overall reflection + transmission matrix $(\beta + \gamma)$, that is,

$$B^+(t/2) = (\gamma + \beta) B^-(t/2) ,$$

and

$$\gamma + \beta = \left(\beta_\infty + \gamma_\infty(t)\right)\left(I + \beta_\infty \gamma_\infty(t)\right)^{-1}$$

$$= \left(X F^{-1} + Y \phi^-(t)\right)\left(Y + X F^{-1} \phi^-(t)\right)^{-1} \quad . \tag{23}$$

4. ARBITRARY SOURCE DISTRIBUTION IN AN INFINITE HALF PLANE

An infinite half-plane with an arbitrary distribution of sources is a common component in many neutron distribution problems. The general solution for the region $0 \leq x < \infty$ may be obtained by addition of the solution

$$B^+ = -Y \phi^- A \quad , \qquad B^- = -X F^{-1} \phi^- A \ ,$$

to the solution which gives the neutron distribution in the right half of an infinite plane, due to an arbitrary source distribution in the right half plane.

Let a unit plane source be located at x' in an infinite medium. Integrating the equation

$$(DM + P)\, B^+(x) = c\, B^-(x) + S(x)\, \delta$$

across the singularity in S gives

$$M\left(B^+(x'_+) - B^+(x'_-)\right) = \delta \ ,$$

that is,

$$B^+(x'_+) - B^-(x'_+) = B^-(x'_-) - B^+(x'_-) = M^{-1}\delta \ . \tag{24}$$

From these boundary conditions, the solution is found to be as follows:

(a) $x < x'$

$$B^+(x) = X \phi^+(x)\, \phi^-(x')\, (Y F - X)^{-1} M^{-1} \delta$$

$$B^-(x) = Y F \phi^+(x)\, \phi^-(x')\, (Y F - X)^{-1} M^{-1} \delta \ .$$

(b) $x > x'$

$$B^+(x) = Y \phi^-(x)\, \phi^+(x')\, (Y - X F^{-1})^{-1} M^{-1} \delta$$

$$B^-(x) = X F^{-1} \phi^-(x)\, \phi^+(x')\, (Y - X F^{-1})^{-1} M^{-1} \delta \ .$$

From the definition of δ and (15a),

$$(Y F - X)^{-1} M^{-1} \delta = \frac{1}{2(1-c)} \alpha\, (Y F + X)^{-1} \ . \tag{25}$$

If now $YF + X = P$, and the notation

$$\int_0^\infty \phi^- |x - x'|\, S(x')\, dx' \quad ,$$

is used to denote the diagonal matrix with elements

$$\int_0^\infty S(x')\, e^{-\alpha_i |x-x'|}\, dx' \quad , \qquad \text{etc.,}$$

then the complete solution for an arbitrary source can be written

$$B^{+}(x) + B^{-}(x) = \frac{1}{2(1-c)} P \int_{0}^{\infty} \phi^{-}|x-x'| S(x') dx' \alpha P^{-1}\delta - P F^{-1}\phi^{-}(x) A \tag{26a}$$

$$B^{+}(0) = \frac{1}{2(1-c)} X \int_{0}^{\infty} \phi^{-}(x') S(x') dx' \alpha P^{-1} \delta - Y \phi^{-}(x) A \tag{26b}$$

$$B^{-}(0) = \frac{1}{2(1-c)} Y F \int_{0}^{\infty} \phi^{-}(x') S(x') dx' \alpha P^{-1}\delta - X F^{-1}\phi^{-1}(x) A \quad . \tag{26c}$$

For comparison, diffusion theory, with

$$\kappa^2 = \Sigma_a / D \, \Sigma_t^2 \quad ,$$

predicts

$$\Phi(x) = \frac{\kappa}{2(1-c)} \int_{0}^{\infty} S(x') e^{-\kappa|x-x'|} dx' - A e^{-\kappa x} \quad . \tag{27}$$

Of particular interest are black body or vacuum boundary conditions at $x = 0$, that is, $B^{+}(0) = 0$. Thus from (26b)

$$A = \frac{1}{2(1-c)} Y^{-1} X \int_{0}^{\infty} \phi^{-}(x') S(x') dx' \alpha P^{-1}\delta \quad . \tag{28}$$

If $Y^{-1}X$ be denoted by Q, then the flux distribution, as given by the double P_N transport theory solution to this problem is

$$\Phi(x) = \frac{1}{2(1-c)} \left[\sum_{j=1}^{N+1} P_{1j} \alpha_j (P^{-1})_{j1} \int_{0}^{\infty} S(x') e^{-\alpha_j |x-x'|} dx' - \sum_{j=1}^{N+1} \sum_{i=1}^{N+1} P_{1i} F_i^{-1} e^{-\alpha_i x} Q_{ij} \alpha_j (P^{-1})_{j1} \int_{0}^{\infty} S(x') e^{-\alpha_j x'} dx' \right] \quad . \tag{29}$$

From this solution the asymptotic component may be extracted for evaluation of a linear extrapolation distance defined by

$$\lambda = \Phi_{asy}(0) \Big/ \Phi'_{asy}(0) \quad .$$

Two cases are worth noting

(a) <u>Plane source at $x' >> 1$</u> (strength S)

Double P_N gives

$$\Phi_{asy}(x) = \frac{\alpha_{11} P_{11} (P^{-1})_{11}}{2(1-c)} \left\{ e^{\alpha_1 x} - F_1^{-1} Q_{11} e^{-\alpha_1 x} \right\} S e^{-\alpha_1 x} \quad , \tag{30a}$$

while diffusion theory, using the relevant linear extrapolation distance for this problem, λ_a, gives

$$\Phi(x) = \frac{\kappa}{2(1-c)} \left\{ e^{\kappa x} - \left(\frac{1-\kappa\lambda_a}{1+\kappa\lambda_a} \right) e^{-\kappa x} \right\} S e^{-\kappa x'} \quad . \tag{30b}$$

(b) Uniform source (strength S per unit length)

Double P_N gives

$$\Phi_{asy}(x) = \frac{S}{2(1-c)}\left\{2 - P_{11}(P^{-1})_{11}\, e^{-\alpha_1 x} + e^{-\alpha_1 x}\sum_{j=1}^{N+1} P_{11}\, F_1^{-1}\, Q_{1j}\,(P^{-1})_{j1}\right\} , \qquad (31a)$$

while diffusion theory gives

$$\Phi(x) = \frac{S}{2(1-c)}\left\{2 - \left(\frac{2}{1+\kappa\lambda_b}\right) e^{-\kappa x}\right\} . \qquad (31b)$$

Thus if $\alpha_1 \approx \kappa$ and $P_{11}(P^{-1})_{11} \approx 1$, which occurs if $(1-c) << 1$, both problems predict the same linear extrapolation distance

$$\kappa\lambda_a = \kappa\lambda_b = \alpha_1\lambda = \frac{1 - F_1^{-1}\, Q_{11}}{1 + F_1^{-1}\, Q_{11}} . \qquad (32)$$

Note that (30b) and (31b) are particular cases of the diffusion theory result for black body boundary conditions,

$$\Phi(x) = \frac{\kappa}{2(1-c)}\left\{\int_0^\infty S(x')\, e^{-\kappa|x-x'|}dx' - e^{-\kappa x}\left(\frac{1-\kappa\lambda}{1+\kappa\lambda}\right)\int_0^\infty S(x')\, e^{-\kappa x'}dx'\right\} . \qquad (33)$$

5. FORMULAE FOR THE DOUBLE P_2 AND P_3 APPROXIMATIONS

The general form of the X and Y matrices is as follows:

$$X = \begin{bmatrix} 1 & 1 & 1 & \cdot & \cdot & \cdot \\ G_1(\alpha_1) & G_1(\alpha_2) & G_1(\alpha_3) & \cdot & \cdot & \cdot \\ G_2(\alpha_1) & G_2(\alpha_2) & G_2(\alpha_3) & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \end{bmatrix} \qquad (34)$$

$$Y = \begin{bmatrix} 1 & 1 & 1 & \cdot & \cdot & \cdot \\ G_1(-\alpha_1) & G_1(-\alpha_2) & G_1(-\alpha_3) & \cdot & \cdot & \cdot \\ G_2(-\alpha_1) & G_2(-\alpha_2) & G_2(-\alpha_3) & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \end{bmatrix} . \qquad (35)$$

Thus $\quad G_1(\alpha_1) = A_1^+ / A_1^0 \quad$ for the $e^{\alpha_1 x}$ solutions,

and $\quad G_1(-\alpha_1) = A_1^+ / A_1^0 \quad$ for the $e^{-\alpha_1 x}$ solutions, etc.

The first of the double P_N equations relates A_0^- and A_0^+ so the elements of the F matrix are

$$F_i = \frac{1}{c}\left\{(2-c) + \alpha_i\left(1 + G_1(\alpha_i)\right)\right\} \qquad (36)$$

5.1 Double P_2, $(0 < c < 1)$

The characteristic equation is

$$\alpha^6 - 12\,\alpha^4(7-2c) + 20\,\alpha^2\left(21 - \frac{43\,c}{3}\right) - 400\,(1-c) = 0\,, \tag{37}$$

and the G functions are

$$G_1(\alpha) = \frac{-5\,\alpha(2+\alpha)}{15\,(2+\alpha)^2 - 4\,\alpha^2}\,, \tag{38a}$$

$$G_2(\alpha) = \frac{2\,\alpha^2}{15\,(2+\alpha)^2 - 4\,\alpha^2}\,. \tag{38b}$$

Note that for $(1-c) << 1$

$$\alpha_1^2 \approx 3\,(1-c)\,\{1 - 0.8\,(1-c)\}\,,$$

which is the correct diffusion theory approximation to κ^2.

5.2 Double P_2, $(c = 0)$

$$(\alpha^2 - 4)\,(\alpha^4 - 80\,\alpha^2 + 100) = 0\,, \tag{39}$$

thus $\quad \alpha_1 = 5 - \sqrt{15}\,, \quad \alpha_2 = 2\,, \quad \alpha^3 = 5 + \sqrt{15}\,.$

In this particular case, there is a simpler form for Y, that is,

$$Y = \begin{bmatrix} 1 & 1 & 1 \\ \frac{2-\alpha_1}{\alpha_1} & 0 & \frac{2-\alpha_3}{\alpha_3} \\ 0.4 & -0.5 & 0.4 \end{bmatrix}\,. \tag{40}$$

5.3 Double P_3 $(0 < c < 1)$

The characteristic equation is

$$\alpha^8 + 20\,\alpha^6\,(-11 + 2c) + 20\,\alpha^4\left(132 - \frac{193}{3}\,c\right) + $$

$$700\,\alpha^2\left(-10 + \frac{23}{3}\,c\right) + 4900\,(1-c) = 0\,, \tag{41}$$

and the G functions are

$$G_1(\alpha) = \frac{\alpha\,\{9\,\alpha^2 - 35\,(2+\alpha)^2\}}{(2+\alpha)\,\{105\,(2+\alpha)^2 - 55\,\alpha^2\}}\,, \tag{42a}$$

$$G_2(\alpha) = \frac{14\,\alpha^2}{105\,(2+\alpha)^2 - 55\,\alpha^2}\,, \tag{42b}$$

$$G_3(\alpha) = \frac{-6\,\alpha^3}{(2+\alpha)\,\{105\,(2+\alpha)^2 - 55\,\alpha^2\}}\,. \tag{42c}$$

5.4 Double P_3 (c = 0)

No special case arises as in the double P_2 for which $\alpha_2 = 2.0$ but the characteristic equation can be written

$$\alpha^4 - 20\,\alpha^3 + 90\,\alpha^2 - 140\alpha + 70 = 0\ , \tag{43}$$

while the j'th column of Y is

$$\left\{ 1 \quad \frac{2-\alpha_j}{\alpha_j} \quad \frac{6-6\,\alpha_j+\alpha_j^2}{\alpha_j^2} \quad \frac{20-30\,\alpha_j+12\,\alpha_j^2-\alpha_j^3}{\alpha_j^3} \right\} \ . \tag{44}$$

The roots in the N = 2 approximation are given by Ziering and Schiff, but are reproduced here, for completeness, in Table 1, which also includes the roots for N = 3.

6. ILLUSTRATIVE APPLICATIONS

The applications given here are intended to illustrate the method of utilizing reflection and transmission matrices and the various functions developed in the general theory. The problems chosen are those for which exact solutions are known or accurate numerical solutions have been tabulated. The double P_N numerical results quoted have been obtained with digital computer programmes utilizing basic sub-routines for performing the matrix operations. For N = 2 and 3, only 3rd and 4th order matrices are involved.

6.1 Slab Escape Probability

A slab of half thickness t, in mean free paths, has a uniformly distributed source. The origin of x is at the slab centre.

6.1.1 c = 0

$$B^+(x) = \frac{1}{2} S\ \delta + Y\ \phi^- A$$

$$B^-(x) = \frac{1}{2} S\ \delta + Y\ \phi^+ A \ .$$

From the boundary conditions $B^-(t) = B^+(-t) = 0$,

$$A = -\frac{1}{2}\ \phi^-(t)\ Y^{-1}\ \delta\ S\ .$$

The average escape probability P_{esc} is therefore

$$P_{esc} = \frac{1}{2\,St}\left(B_0^+(t) + B_1^+(t)\right)$$

$$= \frac{1}{4t}\ \left\{\left(I - Y\ \phi^-(2t)\ Y^{-1}\right)\ \delta\ \right\}_{1+2}\ . \tag{45}$$

6.1.2 0 < c < 1

From symmetry,

$$B^+(x) = \frac{S}{2(1-c)}\ \delta + (X\,\phi^+ + Y\,F\,\phi^-)\ A$$

$$B^-(x) = \frac{S}{2(1-c)}\ \delta + (Y\,F\,\phi^+ + X\,\phi^-)\ A \ .$$

In the same way as for the derivation of (45),

$$P_{esc} = \frac{1}{4t(1-c)}\ [\{I - \left(X\,\phi^+(t) + Y\,F\ \phi^-(t)\right)\left(X\,\phi^-(t) + Y\,F\ \phi^+(t)\right)^{-1}\}\ \delta\ ]_{1+2}\ . \tag{46}$$

Results for a pure absorber, $c = 0$, are given in Table 2 for comparison with the exact values from Case, de Hoffman, and Placzek (1953).

For a slab that scatters neutrons, the usual approximation for average escape probability is

$$P_{esc} = P_{esc}(c=0) \Big/ \left\{1 - c\left(1 - P_{esc}(c=0)\right)\right\} . \tag{47}$$

Numerical results from the double P_3 approximation are compared with results using (47) in Table 3.

6.2 The Milne Problem

An infinite half-plane occupies the space $0 \le x < \infty$, with a source at $+\infty$. The quantities of interest are the angular distribution of the emergent neutrons, that is $B^-(0)$, the linear extrapolation distance λ, the extrapolated end point $|x_0|$, and the ratio of asymptotic flux to transport theory flux at $x = 0$. This problem has been solved essentially, in the double P_N approximation, in Section 4, but the derivation here is to illustrate the general technique of solving slab problems. Only $0 < c < 1$ need be considered.

The form of the solution is

$$B^+(x) = X\phi^+\delta + Y\phi^- A$$

$$B^-(x) = YF\phi^+\delta + XF^{-1}\phi^- A .$$

The boundary condition, $B^+(0) = 0$ gives

$$A = -Y^{-1}X\delta = -Q\delta ,$$

and the emergent neutron flux has the angular distribution

$$B^-(0) = (YF - XF^{-1}Y^{-1}X)\,\delta . \tag{48}$$

The flux distribution comes from

$$B^+(x) + B^-(x) = \left(P\phi^+(x) - PF^{-1}\phi^-(x)\,Q\right)\delta , \tag{49}$$

and the transport theory flux at the surface is

$$\Phi(0) = B_0^- = (YF)_{11} - \sum_{j=1}^{N+1} X_{1j}\, F_j^{-1} Q_{j1} . \tag{50}$$

From (49), the asymptotic flux is

$$\Phi_{asy}(x) = P_{11}\, e^{\alpha_1 x} - P_{11} F_1^{-1} Q_{11}\, e^{-\alpha_1 x} , \tag{51}$$

which gives an extrapolated end point at $x = -x_0$ with

$$x_0 = \frac{-1}{2\alpha_1} \ln (F_1^{-1} Q_{11}) . \tag{52}$$

The linear extrapolation distance λ is given by

$$\lambda = \frac{\Phi_{asy}(0)}{\Phi'_{asy}(0)} = \frac{1 - F_1^{-1} Q_{11}}{\alpha_1(1 + F_1^{-1} Q_{11})} , \tag{53}$$

which agrees with (32).

Table 4 compares the exact results of Case, de Hoffman, and Placzek (1953), with the results of the double P_2 and double P_3 approximations.

6.3 Single Slab Self-Shielding

In this problem a slab of thickness 2t, parameter c_s, is surrounded by an infinite medium, parameter c_m, in which is located a uniformly distributed source S. It is required to calculate the mean flux in the slab, the asymptotic flux depression at the surface, and the linear extrapolation distance at the surface. Suffixes s and m refer to slab and surrounding medium respectively.

In the slab, with origin at the centre, the solution is, from symmetry,

$$B_s^+(x) = \left(X_s \phi_s^+(x) + Y_s F_s \phi_s^-(x) \right) A_s$$

$$B_s^-(x) = \left(Y_s F_s \phi_s^+(x) + X_s \phi_s^-(x) \right) A_s \quad , \tag{54}$$

and therefore

$$B_s^+(x) + B_s^-(x) = 2 (X + YF)_s \cosh(\alpha_s x_s) A_s \quad . \tag{55}$$

In the right half-medium, with origin at the interface, the solution can be written as

$$\left.\begin{aligned} B_m^+(x) &= \frac{S}{2(1-c_m)} \delta + b_m^+(x) \\ B_m^-(x) &= \frac{S}{2(1-c_m)} \delta + b_m^-(x) \end{aligned}\right\} \quad , \tag{56}$$

with

$$\left.\begin{aligned} b_m^+(x) &= Y_m \phi_m^-(x) A_m \\ b_m^-(x) &= X_m F_m^{-1} \phi_m^-(x) A_m \end{aligned}\right\} \quad . \tag{57}$$

The flux at infinity is $\Phi_\infty = S/(1-c_m)$, and at the surface

$$b_m^-(0) = (\beta_\infty)_m b_m^+(0) \quad , \tag{58}$$

with β_∞ as given by (17b).

The overall slab reflection + transmission matrix is

$$\beta_s = (\beta + \gamma)_s \quad ,$$

as given in (23) with (t) replaced by (2t).

For continuity of the Legendre components of the angular flux at the interface,

$$B_s^-(t) = \frac{S}{2(1-c_m)} \delta + (\beta_\infty)_m b_m^+(0)$$

$$\beta_s B_s^-(t) = \frac{S}{2(1-c_m)} \delta + b_m^+(0) \quad . \tag{59}$$

Therefore

$$B_s^-(t) = \left(I - (\beta_\infty)_m \beta_s \right)^{-1} \left(I - (\beta_\infty)_m \right) \frac{S}{2(1-c_m)} \delta \quad , \tag{60}$$

and the slab flux distribution is determined by

$$A_s = \left(Y_s F_s \phi_s^+(t) + X_s \phi_s^-(t) \right)^{-1} B_s^-(t) \quad . \tag{61}$$

The average slab flux may be determined by integration or by neutron conservation. For the latter, the net current into the slab at $x_s = t$ requires

$$\frac{1}{2} \left(B_s^-(t) - B_s^+(t) \right) = \frac{1}{2} (I - \beta_s) B_s^-(t) \quad ,$$

so the average flux $\overline{\Phi}_s$ is

$$\overline{\Phi}_s = \frac{1}{2t(1-c_s)} \{ (I - \beta_s) B_s^-(t) \}_{1+2} \quad . \tag{62}$$

Integration gives a different expression, but equality may be demonstrated by the use of (15a). For the external medium,

$$b_m^+(0) = \beta_s B_s^-(t) - \frac{S}{2(1-c_m)} \delta ,$$

and therefore from (57)

$$A_m = -Y_m^{-1}\left(\frac{S}{2(1-c_m)}\delta - \beta_s B_s^-(t)\right) . \qquad (63)$$

The flux distribution in the external medium is now obtained from (56) and (57) as

$$\Phi_m(x) = \left\{\frac{S}{(1-c_m)}\delta - (Y_m + X_m F_m^{-1})\,\phi_m^-(x)\, Y_m^{-1}\left(\frac{S}{2(1-c_m)}\delta - \beta_s B_s^-(t)\right)\right\}_1 . \qquad (64)$$

From (64) the α_1 component and the Φ_∞ term give the asymptotic flux distribution in the external medium, and this can be written as

$$\left(\Phi_{asy}(x)\right)_m = \frac{S}{1-c_m}\left(1 - g\, e^{-(\alpha_1 x)_m}\right) .$$

The linear extrapolation distance at the surfaceis therefore

$$\lambda = \frac{1-g}{(\alpha_1)_m\, g} . \qquad (65)$$

When $c_s = 0$, that is, a pure absorber, the only modification to the above is in β_s which is now

$$\beta_s = \left(\gamma_\infty(2t)\right)_s .$$

This example clearly shows how any array of slabs, as in a slab lattice, may be analyzed in the double P_N approximation, by the use of computer programmes based on sub-routines for the manipulation of comparatively low order matrices.

For $c_s = 0$, Jeffrey (1960) has published the results of ordinary P_{15} calculations on grey slabs. In Table 5 the results of double P_N calculations are compared with Jeffrey's results. Although these latter results may be taken to be quite accurate, they are presented graphically and are difficult to read, so random errors are present in the quoted values.

7. DISCUSSION OF RESULTS

The lowest eigenvalue $\alpha(1)$, computed from the double P_2 and P_3 approximations, may be compared with the exact value κ given by the equation

$$c = \kappa \tanh^{-1}\kappa$$

c	κ	$\alpha(1)$ (P_2)	$\alpha(1)$ (P_3)
0	1.0000	1.1270	1.0746
0.4	0.9856	1.0255	1.0037
0.5	0.9575	0.9793	0.9654
0.6	0.9073	0.9173	0.9100
0.7	0.8286	0.8321	0.8292
0.8	0.7104	0.7112	0.7105
0.9	0.5254	0.5255	0.5254
0.99	0.1725	0.1725	0.1725

For P_2 the error in κ is less than 1 per cent. for $c > 0.6$, and for P_3 the rangeis $c > 0.5$. In reactor analysis using diffusion theory, $c > 0.99$, so application to strong absorbers in a weakly absorbing medium can be expected to give the correct asymptotic, that is, diffusion theoretic, flux distributions.

Milne's problem is the classical test for transport theory solutions, and the results of Table 4 show that for $c \geq 0.7$, the double P_3 solutions for the significant parameters are remarkably accurate.

This is to be expected as the double P_n method is specifically designed to cope with problems that involve discontinuities in the angular flux. For $c \geq 0.9$ the double P_2 approximation is probably quite adequate.

The slab escape probability results show that for $c = 0$, the double P_3 approximation is very good. The errors in the approximation of Equation 47 for $c \neq 0$ increase with increasing thickness, owing to the breakdown of the assumption that the collision density after multiple scattering is still flat.

The comparison presented in Table 5 is difficult to analyse, owing to uncertainty in the values read from Jeffrey's curves. For $c_m = 0.8$, there appear to be no significant differences. For $c_m = 0.99$, Jeffrey's results, for the linear extrapolation distance only, tend to differ from the double P_n approximations as the thickness of the purely absorbing slab increases. Increasing slab thickness, however, tends to vacuum boundary conditions, which for $c_m = 0.99$ gives the same extrapolation distance as in Milne's problem, and the double P_2 and P_3 approximations have been shown to be practically exact. The excellent agreement between the P_2 and P_3 results that disagree with Jeffrey supports the conclusion that his method does not adequately cope with these problems. Jeffrey used the ordinary P_{15} approximation, and Weinberg and Wigner (1958) show graphically that for Milne's problem with $c = 1$, while the total flux near the boundary is exact, the angular distribution of the emergent flux is still significantly in error.

The general conclusion from the numerical studies must be that for all practical purposes, the double P_3 is to be recommended as an accurate approximation for the problems of slab reactor theory involving strong absorbers adjacent to weak absorbers.

It should be noted that no great attention has been given to the problem of round-off errors in the computer programmes. Thus matrix inversion is done by partitioning into second-order matrices and the results are probably only correct to 5 places. However the agreement to four significant figures actually obtained, as in Milne's problem and slab escape probability, suggests that this is no cause for alarm in most practical problems.

8. REFERENCES

Case, K.M., de Hoffmann, F., and Placzek, G. (1953). – Introduction to the Theory of Neutron Diffusion, Washington D.C. U.S. Govt. Printing Office.

Jeffrey, A. (1960). – Some Properties of Transport Flux and Current at the Surface of Grey Plates. Journal of Nuclear Energy, A., 12:93–101.

Weinberg, A., and Wigner, E. (1958). – The Physical Theory of Neutron Chain Reactors, p. 262. The University of Chicago Press.

Ziering, S., and Schiff, D. (1958). – Yvon's Method for Slabs. Nuclear Science and Engineering, 3:635–647.

9. NOTATION

B	Vector of Legendre coefficients in half-range expansion of angular flux
c	Number of secondaries per collision
C,P,M,	Square matrices in statement of formulation of simultaneous equations
F	Diagonal matrix coupling B^- and B^+ components
$G(\alpha)$	Functions appearing in the definition of X and Y
N	Order of the double P_N approximation
P	Used as above, also to denote the square matrix $(X + YF)$
Q	Used to denote the square matrix $Y^{-1}X$

P_n	n'th order Legendre polynomial
S	Source strength, or source strength per total mean free path
t	Slab dimension in total mean free paths
x	Distance in total mean free paths
X, Y,	Matrices appearing in basic exponential type solution
α_j	Characteristic roots
β_∞	Reflection matrix for infinite half–plane
β	Reflection matrix for finite slab
$\gamma_\infty(t)$	Transmission matrix referring to depth t in an infinite half–plane
γ	Transmission matrix for finite slab
δ	An (N+1)'th order vector with only the leading term non–zero and equal to 1
κ	Inverse diffusion length
λ	Linear extrapolation distance
μ	Cosine of angle between direction vector and x–axis
$\phi(x,\mu)$	Neutron angular flux
$\Phi(x)$	Neutron scalar flux
$\Phi_{asy}(x)$	Asymptotic portion of scalar flux
Suffix s	Denotes slab
m	" surrounding medium
1+2	" sum of first and second components of a vector

TABLE 1

VALUES OF LATENT ROOTS α_j

c	N = 2			N = 3			
	1	2	3	1	2	3	4
0.00	1.12701	2.00000	8.87299	1.07461	1.49256	3.03021	14.4026
0.01	1.12543	1.99554	8.86065	1.07367	1.49011	3.02527	14.3901
0.10	1.10963	1.95434	8.74924	1.06393	1.46699	2.98017	14.2771
0.20	1.08804	1.90631	8.62451	1.04979	1.43895	2.92891	14.15092
0.30	1.06073	1.85616	8.49879	1.03058	1.40864	2.87664	14.0242
0.40	1.02555	1.80431	8.37214	1.00375	1.37668	2.82356	13.8969
0.50	0.979328	1.75153	8.24457	0.965430	1.34425	2.76998	13.7691
0.60	0.917322	1.69898	8.11615	0.909964	1.313081	2.71628	13.6407
0.70	0.832192	1.64811	7.98694	0.829236	1.28503	2.66288	13.5119
0.80	0.711249	1.60054	7.85699	0.710488	1.26134	2.61029	13.3824
0.90	0.525509	1.55766	7.72642	0.525432	1.24224	2.55902	13.2526
0.95	0.379494	1.53825	7.66092	0.379485	1.23427	2.53405	13.1875
0.99	0.172511	1.52376	7.60844	0.172511	1.22856	2.51445	13.1354

TABLE 2

SLAB ESCAPE PROBABILITY ($c = 0$)

2t	Exact	Double P_2	Double P_3
0.20	0.8371	0.8497	0.8413
0.40	0.7403	0.7450	0.7399
0.60	0.6665	0.6670	0.6654
0.80	0.6068	0.6055	0.6060
1.00	0.5568	0.5552	0.5564
1.20	0.5141	0.5126	0.5140
1.60	0.4446	0.4438	0.4447
2.0	0.3903	0.3901	0.3904
3.0	0.2954	0.2957	0.2955
4.0	0.2349	0.2350	0.2349
6.0	0.1637	0.1637	0.1637
8.0	0.1243	0.1243	0.1243
10.0	0.0998	0.0998	0.0998
12.0	0.0833	0.0833	0.0833

TABLE 3

SLAB ESCAPE PROBABILITY ($c \neq 0$)

2t	c	Double P_3	Approx. (47)
0.20	0	0.8413	0.8413
	0.4	0.8983	0.8983
	0.8	0.9636	0.9636
	0.95	0.9907	0.9907
1.00	0	0.5564	0.5564
	0.4	0.6758	0.6764
	0.8	0.8617	0.8625
	0.95	0.9614	0.9617
4.0	0	0.2349	0.2349
	0.4	0.3337	0.3385
	0.8	0.5935	0.6055
	0.95	0.8521	0.8600

TABLE 4

MILNE'S PROBLEM

c	$\Phi_{asy}(0) / \Phi(0)$			x_o			λ		
	Exact	P_2	P_3	Exact	P_2	P_3	Exact	P_2	P_3
0.1	1.0375	1.0273	1.0306	8.539	2.1700	2.4592	1.0000	0.8867	0.9299
0.2	1.0801	1.0560	1.0630	3.926	1.8176	2.0537	0.9993	0.8845	0.9274
0.3	1.1218	1.0857	1.0965	2.497	1.5876	1.7734	0.9889	0.8799	0.9214
0.4	1.1535	1.1157	1.1296	1.826	1.4058	1.5419	0.9606	0.8717	0.9100
0.5	1.1763	1.1449	1.1599	1.441	1.2502	1.3393	0.9201	0.8587	0.8907
0.6	1.1932	1.1716	1.1849	1.192	1.1124	1.1618	0.8750	0.8394	0.8622
0.7	1.2061	1.1943	1.2032	1.018	0.9896	1.0112	0.8300	0.8135	0.8261
0.8	1.2161	1.2115	1.2155	0.889	0.8815	0.8881	0.7871	0.7817	0.7864
0.9	1.2241	1.2233	1.2240	0.789	0.7886	0.7895	0.7472	0.7464	0.7471
0.99		1.2301	1.2300		0.7177	0.7176		0.7141	0.7140
1.0	1.2305			0.7104			0.7104		

TABLE 5

SLAB SELF-SHIELDING ($c_s = 0$)

	$2t$	$c_m = 0.8$			$c_m = 0.9$			$c_m = 0.95$			$c_m = 0.99$		
		J*	P_2	P_3	J	P_2	P_3	J	P_2	P_3	J	P_2	P_3
λ	0.01	98	97.8	98.9	80	80.4	80.6	73	73.5	73.7	68	68.8	69.0
	0.02	51	49.5	50.2	42	40.7	40.9	38	37.3	37.4	35	34.9	35.0
	0.10	11	10.87	11.0	9	8.98	9.6	8.1	8.23	8.30	7.6	7.71	7.71
	0.20	6	5.93	6.02	4.9	4.91	4.93	4.5	4.50	4.51	4.2	4.21	4.22
	0.50	2.85	2.85	2.88	2.35	2.34	2.34	2.1	2.14	2.13	1.98	1.99	1.98
	1.0	1.8	1.80	1.83	1.45	1.46	1.46	1.3	1.32	1.32	1.2	1.21	1.21
	2.0	1.35	1.34	1.36	1.10	1.08	1.07	0.95	0.963	0.963	0.83	0.867	0.866
	3.0	1.25	1.23	1.25	1.0	0.99	0.99	0.88	0.882	0.882	0.75	0.790	0.789
$\left.\frac{\Phi_{asy}(0)}{\Phi_\infty}\right\|_m$	0.20	0.81	0.808	0.810	0.71	0.720	0.721	0.63	0.630	0.631	0.41	0.420	0.421
	0.50	0.67	0.670	0.672	0.55	0.552	0.551	0.44	0.448	0.447	0.25	0.255	0.254
	1.0	0.56	0.561	0.565	0.44	0.434	0.435	0.33	0.334	0.334	0.16	0.172	0.172
	2.0	0.49	0.487	0.492	0.36	0.361	0.361	0.27	0.267	0.267	0.12	0.130	0.130
	3.0	0.47	0.467	0.472	0.34	0.342	0.342	0.25	0.250	0.250	0.11	0.120	0.119
	4.0	0.46	0.461	0.466	0.33	0.336	0.337	0.24	0.245	0.246	0.11	0.117	0.116
		$c_m = 0.85$			$c_m = 0.9$			$c_m = 0.95$			$c_m = 0.99$		
$\frac{\bar{\Phi}_s}{\Phi_\infty}$	0.20	0.64	0.606	0.603	0.60	0.569	0.567	0.53	0.502	0.501	0.35	0.337	0.337
	0.50	0.41	0.396	0.397	0.36	0.359	0.360	0.32	0.298	0.298	0.17	0.173	0.173
	1.0	0.26	0.253	0.253	0.23	0.224	0.224	0.18	0.178	0.178	0.10	0.0962	0.0962
	2.0	0.15	0.143	0.143	0.13	0.125	0.125	0.10	0.0978	0.0978	0.05	0.0505	0.0505
	3.0	0.11	0.0987	0.0987	0.09	0.0860	0.0860	0.07	0.0666	0.0666	0.033	0.0341	0.0340

* Jeffrey's results.

APPENDIX 1

DOUBLE P_3 FORTRAN PROGRAMMES AND SUBROUTINES

The following programmes have been used to make the calculations quoted. All the basic matrix operations are performed by a set of subprogrammes which are therefore available for further applications of the method.

Subroutine Name	Operation	Dimensions	Remarks
SUPA(C,A,L,B)	C = A + B (L = 1) C = A − B (L = 2)	A(4,4), B(4,4) C(4,4)	
SUPU(C,L,A)	C = I − A(L = 1) C = I + A (L = 2)	A(4,4), C(4,4)	I = unit matrix
SUPV(C,A,B)	C = AB	A(4,4), B(4), C(4)	C, B, vectors
SUPM(C,A,B)	C = AB	A(4,4), B(4,4), C(4,4)	Multiplication
SUPT(C,A)	C = A^{-1}	A(4,4), C(4,4)	Inversion by partitioning
SUPDP (C,A,B)	C = AB	A(4), B(4,4)C(4,4)	A diagonal
SUPPD(C,A,B)	C = AB	A(4,4), B(4), C(4,4)	B diagonal
SUPR(C,A,X,Y,F, FM)	A(I) = α_i FM = F^{-1} etc.	A(4), F(4), FM(4) X(4,4), Y(4,4)	Computes basic eigenvalues and matrices, $c \neq 0$
SUPRC(A,Y)	Y and α_i	Y (4,4), A(4)	c = 0

APPENDIX 1

DOUBLE P_3 FORTRAN PROGRAMMES AND SUBROUTINES

```
C       DOUBLE P3 ESCAPE PROBABILITY
        DIMENSION ALP(4),F(4),FM(4),X(4,4),Y(4,4),
       1PP(4),PM(4),WA(4,4),WB(4,4),WC(4,4),WD(4,4)
      8 READ 1,C,T
      1 FORMAT (2F6.2)
        IF(C)3,2,3
      2 CALL SUPRC (ALP,Y)
        CALL SUPT (WA,Y)
        DO 4I=1,4
      4 PM(I)=EXPF(-2.*ALP(I)*T)
        CALL SUPDP(WB,PM,WA)
        CALL SUPM(WD,Y,WB)
        GO TO 5
      3 CALL SUPR(C,ALP,X,Y,F,FM)
        DO 6I=1,4
        PP(I)=EXPF(ALP(I)*T)
      6 PM(I)=1./PP(I)
        CALL SUPPD(WA,X,PM)
        CALL SUPPD(WB,Y,F)
        CALL SUPPD(WC,WB,PP)
        CALL SUPA(WB,WA,1,WC)
        CALL SUPT(WC,WB)
        CALL SUPPD(WB,X,PP)
        CALL SUPPD(WA,Y,F)
        CALL SUPPD(WD,WA,PM)
        CALL SUPA(WA,WB,1,WD)
        CALL SUPM(WD,WA,WC)
      5 CALL SUPU(WA,1,WD)
        PESC=(WA(1,1)+WA(2,1))/(4.*T*(1.-C))
        PUNCH 7,C,T,PESC
      7 FORMAT (2F6.2,E13.4)
        GO TO 8
        END

C       MILNES PROBLEM BY DOUBLE P3
        DIMENSION A(4),X(4,4),Y(4,4),F(4),FM(4),WA(4,4),WB(4,4),WC(4,4)
      4 READ 1,C
      1 FORMAT(F4.2)
        CALL SUPR(C,A,X,Y,F,FM)
        CALL SUPT(WA,Y)
        CALL SUPM(WB,WA,X)
        Z=FM(1)*WB(1,1)
        PASY=X(1,1)+Y(1,1)*F(1)
        PASY=PASY*(1.-Z)
        PT=Y(1,1)*F(1)
        CALL SUPDP(WA,FM,WB)
        CALL SUPM(WC,X,WA)
        PT=PT-WC(1,1)
        PT=PASY/PT
        XQ=-1.*LOGF(Z)/(2.*A(1))
        D=(1.-Z)/(A(1)*(1.+Z))
        PUNCH 2,C,A(1),A(2),A(3),A(4)
      2 FORMAT(F4.2,4E13.4)
        PUNCH 3,C,PT,XQ,D
      3 FORMAT(F4.2,3E13.4)
        GO TO 4
        END
```

APPENDIX 1 (continued)

```
C     SINGLE SLAB SELF SHIELDING BY DOUBLE P3
      DIMENSION AM(4),XM(4,4),YM(4,4),FM(4),FMM(4),
     1AS(4),XS(4,4),YS(4,4),FS(4),FSM(4),PH(4),BM(4,4),
     2BS(4,4),CJM(4),CJS(4),WA(4,4),WB(4,4),WC(4,4),WD(4,4),Z(4)
    1 READ 2,CM,CS,T
    2 FORMAT (3F6.2)
      CALL SUPR(CM,AM,XM,YM,FM,FMM)
      CALL SUPT(WA,YM)
      DO 20I=1,4
   20 Z(I)=WA(1,I)
      CALL SUPDP(WB,FMM,WA)
      CALL SUPM(BM,XM,WB)
      IF(CS) 5,6,5
    6 CALL SUPRC(AS,YS)
      GO TO 7
    5 CALL SUPR(CS,AS,XS,YS,FS,FSM)
    7 DO 3I=1,4
    3 PH(I)=EXPF(-AS(I)*T)
      CALL SUPT(WA,YS)
      CALL SUPPD(WB,YS,PH)
      IF(CS)8,9,8
    9 CALL SUPM(BS,WB,WA)
      GO TO 10
    8 CALL SUPPD (WC, XS, FSM)
      CALL SUPA (WD, WB, 1, WC)
      CALL SUPPD (WA, WC, PH)
      CALL SUPA (WB, Y, 1, WA)
      CALL SUPT (WA, WB)
      CALL SUPM (BS, WD, WA)
   10 CALL SUPM(WA,BM,BS)
      CALL SUPU(WB,1,WA)
      CALL SUPT(WA,WB)
      CALL SUPU(WB,1,BM)
      CALL SUPM(WC,WA,WB)
      DO 11I=1,4
   11 CJS(I)=.5*WC(I,1)
      CALL SUPV(CJM,BS,CJS)
      CJM(1)=CJM(1)-.5
      P=.0
      DO 12I=1,4
   12 P=P+Z(I)*CJM(I)
      P = P*(YM(1,1)+XM(1,1)*FMM(1))
      PASY=1.+P
      EXT=-PASY/(AM(1)*P)
      CALL SUPU(WB,2,BS)
      CALL SUPV(Z,WB,CJS)
      PT=Z(1)
      CALL SUPU(WA,1,BS)
      CALL SUPV(Z,WA,CJS)
      PM=Z(1)+Z(2)
      PM=PM/(T*(1.-CS))
      PUNCH 4, CM,CS,T,PM,PASY,PT,EXT
    4 FORMAT (3F6.2,4E13.4)
      GO TO 1
      END
```

APPENDIX 1 (continued)

```
      SUBROUTINE SUPR(C,A,X,Y,F,FM)
      DIMENSION A(4),X(4,4),Y(4,4),F(4),FM(4)
      CA = 20.*(2.*C-11.)
      CB = 20.*(132.-(193.*C/3.))
      CC = 700.*((23.*C/3.)-10.)
      CD = 4900.*(1.-C)
      A(1) = (1.-C)*(2.5-1.4*(1.-C))
      A(2) = 2.3-.9*C
      A(3) = 9.-2.6*C
      A(4) = 190.
      DO 2 I=1,4
      Q = A(I)
    4 DEL = (((Q+CA)*Q+CB)*Q+CC)*Q+CD
      DEL = DEL/(((4.*Q+3.*CA)*Q+2.*CB)*Q+CC)
      Q = Q-DEL
      IF (DEL) 1,2,3
    1 IF(DEL+Q*1.E-6)4,2,2
    3 IF(DEL-Q*1.E-6)2,2,4
    2 A(I) = SQRTF(Q)
      DO 12 I = 1,4
      L = 1
      P = A(I)
      GO TO 15
   14 P = -A(I)
   15 R = 2.+P
   19 Z = 1./(105.*R*R-55.*P*P)
      AG = Z*P*(9.*P*P-35.*R*R)/R
      AJ = Z*14.*P*P
      AK = -6.*Z*P*P*P/R
      GO TO (16,18),L
   16 X(1,I) = 1.
      X(2,I) = AG
      X(3,I) = AJ
      X(4,I) = AK
      F(I) = (2.-C+P*(1.+AG))/C
      L = 2
      GO TO 14
   18 Y(1,I) = 1.
      Y(2,I) = AG
      Y(3,I) = AJ
      Y(4,I) = AK
      FM(I) = 1./F(I)
   12 CONTINUE
      RETURN
      END

      SUBROUTINE SUPT(C,A)
      DIMENSION A(4,4),P(2,2),Q(2,2),R(2,2),S(2,2),C(4,4)
      X=1./(A(2,1)*A(1,2)-A(1,1)*A(2,2))
      P(1,1)=A(2,2)*X
      P(1,2)=-A(1,2)*X
      P(2,1)=-A(2,1)*X
```

APPENDIX 1 (continued)

```
      P(2,2)=A(1,1)*X
      X=1./(A(4,3)*A(3,4)-A(3,3)*A(4,4))
      Q(1,1)=A(4,4)*X
      Q(1,2)=-A(3,4)*X
      Q(2,1)=-A(4,3)*X
      Q(2,2)=A(3,3)*X
      R(1,1)=P(1,1)*A(1,3)+P(1,2)*A(2,3)
      R(1,2)=P(1,1)*A(1,4)+P(1,2)*A(2,4)
      R(2,1)=P(2,1)*A(1,3)+P(2,2)*A(2,3)
      R(2,2)=P(2,1)*A(1,4)+P(2,2)*A(2,4)
      S(1,1)=Q(1,1)*A(3,1)+Q(1,2)*A(4,1)
      S(1,2)=Q(1,1)*A(3,2)+Q(1,2)*A(4,2)
      S(2,1)=Q(2,1)*A(3,1)+Q(2,2)*A(4,1)
      S(2,2)=Q(2,1)*A(3,2)+Q(2,2)*A(4,2)
      P(1,1)=A(1,1)+A(1,3)*S(1,1)+A(1,4)*S(2,1)
      P(1,2)=A(1,2)+A(1,3)*S(1,2)+A(1,4)*S(2,2)
      P(2,1)=A(2,1)+A(2,3)*S(1,1)+A(2,4)*S(2,1)
      P(2,2)=A(2,2)+A(2,3)*S(1,2)+A(2,4)*S(2,2)
      Q(1,1)=A(3,3)+A(3,1)*R(1,1)+A(3,2)*R(2,1)
      Q(1,2)=A(3,4)+A(3,1)*R(1,2)+A(3,2)*R(2,2)
      Q(2,1)=A(4,3)+A(4,1)*R(1,1)+A(4,2)*R(2,1)
      Q(2,2)=A(4,4)+A(4,1)*R(1,2)+A(4,2)*R(2,2)
      X=1./(P(1,1)*P(2,2)-P(1,2)*P(2,1))
      C(1,1)=X*P(2,2)
      C(1,2)=-X*P(1,2)
      C(2,1)=-X*P(2,1)
      C(2,2)=X*P(1,1)
      X=1./(Q(1,1)*Q(2,2)-Q(1,2)*Q(2,1))
      C(3,3)=X*Q(2,2)
      C(3,4)=-X*Q(1,2)
      C(4,3)=-X*Q(2,1)
      C(4,4)=X*Q(1,1)
      C(1,3)=R(1,1)*C(3,3)+R(1,2)*C(4,3)
      C(1,4)=R(1,1)*C(3,4)+R(1,2)*C(4,4)
      C(2,3)=R(2,1)*C(3,3)+R(2,2)*C(4,3)
      C(2,4)=R(2,1)*C(3,4)+R(2,2)*C(4,4)
      C(3,1)=S(1,1)*C(1,1)+S(1,2)*C(2,1)
      C(3,2)=S(1,1)*C(1,2)+S(1,2)*C(2,2)
      C(4,1)=S(2,1)*C(1,1)+S(2,2)*C(2,1)
      C(4,2)=S(2,1)*C(1,2)+S(2,2)*C(2,2)
      RETURN
      END

      SUBROUTINE SUPDP(C,A,B)
      DIMENSION C(4,4),A(4),B(4,4)
      DO 2I=1,4
      DO 2J=1,4
    2 C(I,J)=A(I)*B(I,J)
      RETURN
      END

      SUBROUTINE SUPM(C,A,B)
```

APPENDIX 1 (continued)

```
      DIMENSION C(4,4),A(4,4),B(4,4)
      DO 2I=1,4
      DO 2J=1,4
      C(I,J)=0.
      DO 2K=1,4
    2 C(I,J)=C(I,J)+A(I,K)*B(K,J)
      RETURN
      END

      SUBROUTINE SUPRC(ALP,Y)
      DIMENSION ALP(4),Y(4,4)
      ALP(1) = 1.07461
      ALP(2) = 1.49255
      ALP(3) = 3.03021
      ALP(4) = 14.4026
      DO 2J=1,4
      X=1./ALP(J)
      Y(1,J)=1.
      Y(2,J)=2.*X-1.
      Y(3,J)=(6.*X-6.)*X+1.
    2 Y(4,J)=((20.*X-30.)*X+12.)*X-1.
      RETURN
      END

      SUBROUTINE SUPPD(C,B,A)
      DIMENSION C(4,4),B(4,4),A(4)
      DO 2I=1,4
      DO 2J=1,4
    2 C(I,J)=B(I,J)*A(J)
      RETURN
      END

      SUBROUTINE SUPA(C,A,L,B)
      DIMENSION C(4,4),A(4,4),B(4,4)
      DO 3I=1,4
      DO 3J=1,4
      GO TO (2,1),L
    2 C(I,J)=A(I,J)+B(I,J)
      GO TO 3
    1 C(I,J)=A(I,J)-B(I,J)
    3 CONTINUE
      RETURN
      END

      SUBROUTINE SUPU(C,L,A)
      DIMENSION C(4,4), A(4,4)
      B=1.
      GO TO (1,2),L
    1 B=-B
    2 DO 4 I = 1,4
      DO4J=1,4
    4 C(I,J)=B*A(I,J)
```

APPENDIX 1 (continued)

```
  C(1,1) = 1.+C(1,1)
  C(2,2) = 1.+C(2,2)
  C(3,3) = 1.+C(3,3)
  C(4,4) = 1.+C(4,4)
  RETURN
  END

  SUBROUTINE SUPV(C,A,B)
  DIMENSION C(4),A(4,4),B(4)
  DO 2I=1,4
  C(I)=0.
  DO 2K=1,4
2 C(I)=C(I)+A(I,K)*B(K)
  RETURN
  END
```